

Newton's cool

This program can be looked upon as a macabre twist of the game of Life. Morbid-minded physicists will take delight in this unusual application of Newton's Law of Cooling. Botanists, biologists and other lay people are let in on the secret first so that they too can play the game.

J.D. Lee
and
T.D. Lee

NEWTON'S Law of Cooling states that the rate at which a body cools in a draught is directly proportional to the excess temperature — that is, the temperature difference between the body and its surroundings. Whilst this should be well known by physicists, who regard any object as a body, it is less frequently known by others.

To illustrate this law an example is chosen which is likely to be remembered by a wide variety of morbid users. The example deals with bodies — dead bodies! The way in which the time of death of a body may be established from temperature readings will be described. This will be immensely useful to potential pathologists and aspiring assassins, and a computer program is provided for the benefit of non-physicists.

Background

When alive, a human body is closely regulated to maintain a temperature of 98.4°F (approximately 37°C) except during illness such as a fever. When a person dies, their body is no longer maintained at this temperature and consequently it gradually cools towards room temperature. For a physicist's type of body, for example a bar of metal, the rate at which the heat is conducted along the bar is given by:

$$-\frac{dQ}{dt} = K A \frac{\Delta\theta}{\Delta x} \quad (1)$$

where $-\frac{dQ}{dt}$ is the rate of heat loss with time,
K is the thermal conductivity of the metal,
A is the (cross sectional) area through which heat travels,
 $\Delta\theta$ is the temperature difference between the two ends,
 Δx is the distance between the two ends.

For a human body, the heat is conducted from the centre of the body, through the skin and clothes to the air. In a strong draught the warmer air is immediately blown away. The constant K in equation (1) represents the thermal conductivity of skin and clothes combined, A is the surface area of the body and Δx is the thickness of skin and clothes. Not only are these three terms unknown, they also vary depending on the physique and state of dress of the particular body.

Nevertheless, they are constant for any one body. Thus:

$$-\frac{dQ}{dt} \text{ is proportional to } \Delta\theta \quad (2)$$

Moreover the heat content, Q, of a body is its heat capacity multiplied by its absolute temperature θ . Thus: Q is proportional to θ , hence

$$-\frac{dQ}{dt} \text{ is proportional to } -\frac{d\theta}{dt} \quad (3)$$

Combining equations (2) and (3) shows that the rate of cooling, $-\frac{d\theta}{dt}$, of the body is proportional to the excess temperature, $\Delta\theta$. Newton arrived at the same conclusion about three hundred years ago!

Programming the macabre!

Mathematically it can be shown that the body temperature falls exponentially towards the air temperature. If a body temperature reading is taken at an unknown time after death, it is not possible to calculate when the body was at 98.4°F since the proportionality constant is not known. However, if two temperature readings are taken with a known time interval between them, then the time of death may be calculated.

$$\text{Time of death} = \frac{\ln \left[\frac{\text{first body temperature} - \text{air temperature}}{\text{body temperature} - \text{air temperature}} \right]}{\ln \left[\frac{\text{second body temperature} - \text{air temperature}}{\text{body temperature} - \text{air temperature}} \right]} \times \text{time between readings}$$

The time of death thus calculated is given as the time before the first temperature reading was taken. Unfortunately Newton's Law of Cooling only applies in a strong constant draught, which would be the case in an exposed windy location, or in an air conditioned building. In still air, the air warms up and natural convection occurs. The rate of cooling $-\frac{d\theta}{dt}$ is given by

$$-\frac{d\theta}{dt} \text{ is proportional to } \Delta\theta^{3/4}$$

rather than

$$-\frac{d\theta}{dt} \text{ is proportional to } \Delta\theta$$

as given by Newton's Law of Cooling. The time of death may be calculated.

$$\text{Time of death} = \frac{\ln \left[\frac{\text{first body temperature} - \text{air temperature}}{\text{body temperature} - \text{air temperature}} \right]}{\ln \left[\frac{\text{second body temperature} - \text{air temperature}}{\text{body temperature} - \text{air temperature}} \right]} \times \text{time between readings}$$

The Five-Fourths Law of Cooling was determined empirically by Dulong and Petit, and justified theoretically by Lorentz in 1881. Users who are surprised at their results are referred to those mentioned above or to Newton himself!

A BASIC program is provided, written in a most elementary sub-set of the language, which should facilitate its implementation on a wide variety of computers. A sample run is also provided.

Description of the program

The program first asks if the user requires full instructions. An answer of YES or NO is expected and all other responses are rejected. Depending on the answer explicit or shortened messages are printed during the first run. Regardless of the answer, short messages are always given on the second and subsequent runs.

The user is invited to choose whether to use the Celsius or Fahrenheit temperature scales. The reply is checked and only C or F are allowed.

In turn the air temperature, the first body temperature and the second body temperature are requested. Checks are performed to ensure that the numbers entered are reasonable. Warning messages are printed if the values are out of range and the user has to re-type an acceptable value. Finally the user is asked for the time of the interval between the temperature readings. This too is checked, and must be positive and less than five hours.

The time of death is calculated using Newton's Law of Cooling (in a draught), and the Five-Fourths Law.

An explanation of the methods is pro-

vided on request and finally the user is asked if he would like another run.

List of variables

The strings Q\$ and I\$ are used for the replies to questions and whether full instructions are required respectively.

These are DIMensioned in line 10 so that I\$ may contain up to three characters and Q\$ up to ten characters. For a number of versions of BASIC strings are handled in a different way and DIM I\$(3) reserves space for four strings I\$(0), I\$(1), I\$(2) and I\$(3). For

such implementations of BASIC line 10 should be omitted.

- A Air temperature surroundings
- B Body temperature (when alive)
- D Death time in minutes before first reading
- F First temperature reading made on corpse
- S Second temperature reading made on corpse
- T Time in minutes between the two readings

Program Listing

```

10 DIM I$(3), Q$(10)
20 PRINT TAB(30); "Time of Death"
30 PRINT TAB(30); "=====
40 PRINT
50 PRINT "Would you like FULL instructions"
60 GOSUB 940
70 LET I$ = Q$
80 IF I$ = "NO" THEN 160
90 PRINT
100 PRINT "This program calculates how long a person has been dead"
110 PRINT "from two body temperature readings, the time between the"
120 PRINT "readings and the surrounding air temperature. Newton's"
130 PRINT "Law of Cooling is assumed if the body is in a draught"
140 PRINT "otherwise the Five Fourths Law of Natural Convection is used"
150 PRINT
160 PRINT "Would you like to work in degrees Celcius or Fahrenheit"
170 IF I$ = "NO" THEN 190
180 PRINT "Type C or F and press RETURN"
190 INPUT Q$
200 REM *** SET NORMAL BODY TEMPERATURE B
210 LET B = 98.6
220 IF Q$ = "F" THEN 270
230 LET B = 37
240 IF Q$ = "C" THEN 270
250 PRINT "Reply '"; Q$; "' not understood. Re-";
260 GOTO 180
270 PRINT "Type the air temperature"
280 INPUT A
290 IF (A + 40) * (A - B) < 0 THEN 330
300 PRINT "The air temperature must be between -40 degrees"
310 PRINT "and"; B; " degrees. Re-";
320 GOTO 270
330 PRINT "Type the first body temperature"
340 INPUT F
350 IF (F - B) * (F - A) < 0 THEN 390
360 PRINT "The first body temperature must be between"; B; " and"; A;
370 PRINT "degrees. Re-";
380 GOTO 330
390 PRINT "Type the second body temperature"
400 INPUT S
410 IF (S - F) * (S - A) < 0 THEN 450
420 PRINT "The second body temperature must be between"; F; " and"; A;
430 PRINT "degrees. Re-";
440 GOTO 390
450 LET S = S - A
460 LET F = F - A
470 LET B = B - A
480 PRINT "Type the time in minutes between temperature readings"
490 IF I$ = "NO" THEN 510
500 PRINT "Then press RETURN"
510 INPUT T
520 IF T * (T - 300) < 0 THEN 570
530 PRINT "The time must be between 0 and 300 minutes (five hours)"
540 PRINT "Re-";
550 GOTO 480
560 REM *** CALCULATE TIME OF DEATH USING NEWTON'S LAW OF COOLING
570 LET D = INT((LOG(F / B) * T / LOG(S / F) + 0.5))
580 PRINT "Assuming that the body was in a strong constant wind,"
590 PRINT "the person died";
600 IF D < 60 THEN 620
610 PRINT INT(D / 60); " hours and";
620 PRINT D - 60 * INT(D / 60); " minutes before the first reading."
630 PRINT
640 REM CALCULATE TIME OF DEATH USING FIVE FOURTHS LAW
650 LET D = INT((B^(-.25) - F^(-.25)) * T / (F^(-.25) - S^(-.25)) + 0.5)
660 PRINT "If the body was in still air then a better estimate is"
670 IF D < 60 THEN 690
680 PRINT INT(D / 60); " hours and";
690 PRINT D - 60 * INT(D / 60); " minutes before the first reading."
700 PRINT
710 PRINT "Would you like an explanation of the methods"
720 GOSUB 930
730 IF Q$ = "NO" THEN 850
740 PRINT
750 PRINT "The first method uses Newton's Law of Cooling which assumes"
760 PRINT "that the rate of cooling of a body is proportional to the"

```

```

770 PRINT "temperature difference between the body and the atmosphere."
780 PRINT "Newton's Law applies if the body is in a strong constant"
790 PRINT "draught eg. an air conditioned room. Such cooling is called"
800 PRINT "FORCED convection. If the atmosphere is still Newton's Law"
810 PRINT "does not apply and the heat loss is proportional to the"
820 PRINT "excess temperature to the power 1.25. This is called the"
830 PRINT "Five Fourths Law for NATURAL convection and gives rise to"
840 PRINT "the second result."
850 PRINT
860 PRINT "Would you like another run"
870 GOSUB 930
880 LET I$ = "NO"
890 IF Q$ = "YES" THEN 150
900 PRINT "You are finished - Rigor Mortis has set in"
910 STOP
920 REM *** SUBROUTINE TO SORT OUT YES / NO ANSWERS
930 IF I$ = "NO" THEN 950
940 PRINT "Type YES or NO and press RETURN"
950 INPUT Q$
960 IF Q$ = "YES" THEN 1000
970 IF Q$ = "NO" THEN 1000
980 PRINT "Reply '"; Q$; "' not understood. Re-";
990 GOTO 940
1000 RETURN
1010 END

```

Sample run

```

Time of Death
=====
Would you like FULL instructions
Type YES or NO and press RETURN
? YES

This program calculates how long a person has been dead
from two body temperature readings, the time between the
readings and the surrounding air temperature. Newton's
Law of Cooling is assumed if the body is in a draught
otherwise the Five Fourths Law of Natural Convection is used

Would you like to work in degrees Celcius or Fahrenheit
Type C or F and press RETURN
? C
Type the air temperature
? 6
Type the first body temperature
? 25
Type the second body temperature
? 14
Type the time in minutes between temperature readings
Then press RETURN
? 45
Assuming that the body was in a strong constant wind,
the person died 25 minutes before the first reading.

If the body was in still air then a better estimate is
21 minutes before the first reading.

Would you like an explanation of the methods
Type YES or NO and press RETURN
? YES

The first method uses Newton's Law of Cooling which assumes
that the rate of cooling of a body is proportional to the
temperature difference between the body and the atmosphere.
Newton's Law applies if the body is in a strong constant
draught eg. an air conditioned room. Such cooling is called
FORCED convection. If the atmosphere is still Newton's Law
does not apply and the heat loss is proportional to the
excess temperature to the power 1.25. This is called the
Five Fourths Law for NATURAL convection and gives rise to
the second result.

Would you like another run
Type YES or NO and press RETURN
? NO
You are finished - Rigor Mortis has set in
OK,

```